
**************************************************************************************************************
**************************************************************************************************************
* Sintaxis para la ENADID 1992.
* Datos Generales

* Primero sacamos la frecuencia de la variable sexo, sin ponderar para ver el número de observaciones.

FREQUENCIES VARIABLES=P3_7
  /ORDER=ANALYSIS.

VALUE LABELS P3_7 1 'Hombres' 2 'Mujeres'.

* Se tienen 277,552 observaciones. Ahora procedemos a expandir la base de datos

weight by FVIV.

FREQUENCIES VARIABLES=P3_7
  /ORDER=ANALYSIS.

weight off.

* Una vez expandida la base de datos tenemos 85966216 observaciones. Sin embargo, tenemos que reescalar la base de datos, y 
para ello calculamos el ponderador reescalado

COMPUTE pond_datos_gral_1992=FVIV* (277552/85966216).
EXECUTE.

weight by pond_datos_gral_1992.

FREQUENCIES VARIABLES=P3_7
  /ORDER=ANALYSIS.

* Ya que tenemos la base con el ponderador adecuado podemos ver la distribución por edad de la población en 1992 según la 
ENADID

RECODE
  P3_8
(0=0)
(1 thru 4=1)
(5 thru 9=2)
(10 thru 14=3) 
(15 thru 19=4)
(20 thru 24=5)
(25 thru 29=6)
(30 thru 34=7) 
(35 thru 39=8)
(40 thru 44=9)
(45 thru 49=10)
(50 thru 54=11)
(55 thru 59=12)
(60 thru 64=13) 
(65 thru 69=14)
(70 thru 74=15)
(75 thru 79=16)
(80 thru 84=17)
(85 thru 89=18)
(90 thru 94=19)
(95 thru 98=20)
(99=99)
INTO edad_quinquenal.
VARIABLE LABELS  edad_quinquenal 'Edad quinquenal'.
VALUE LABELS edad_quinquenal 0 'Menores 1 año' 1 '1-4' 2 '5-9' 3 '10-14' 4 '15-19' 5 '20-24' 6 '25-29' 7 '30-34' 8 '35-39' 9 '40-44' 
10 '45-49' 11 '50-54' 12 '55-59' 
13 '60-64' 14 '65-69' 15 '70-74' 16 '75-79' 17 '80-84' 18 '85-89' 19 '90-94' 20 '95-98',99 "N.E.".
EXECUTE.

FREQUENCIES VARIABLES=edad_quinquenal
  /ORDER=ANALYSIS.

* Otra caracrerística que se puede observar es el estado civil de las personas mayores de 18 años.

USE ALL.
COMPUTE filter_$=(P3_8 >= 18).



VARIABLE LABEL filter_$ 'P3_8 >= 18 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMAT filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.

Value labels P6_1 1 'Union libre' 2 'Separado' 3 'Divorciado' 4 'Viudo' 5 'Casado' 6 'Soltero' 9 'N.E.' .

FREQUENCIES VARIABLES=P6_1
  /ORDER=ANALYSIS.

USE ALL.

**************************************************************************************************************
**************************************************************************************************************
* Sintaxis para la ENADID 1992.
* Anticoncepción

* Al igual que en la parte del inicio, tenemos que ver el número de casos sin ponderar. Utilizamos la variable entidad puesto que no 
hay una variable más sencilla en la base.

FREQUENCIES VARIABLES=IENT
  /ORDER=ANALYSIS.

* Vemos que hay 69538 casos en esta base de datos de anticoncepción. 

weight by FMUJ.

FREQUENCIES VARIABLES=IENT
  /ORDER=ANALYSIS.

* Una vez que ponderamos hay 23497466 casos, entonces procedemos a reescalar la base.

weight off.

COMPUTE pond_antic_1992=FMUJ* (69538/23497466).
EXECUTE.

weight by pond_antic_1992.

FREQUENCIES VARIABLES=IENT
  /ORDER=ANALYSIS.

* Y vemos que tenemos ya el mismo numero de casos que originalmente se tenían. 

* Alguna vez ha usado algún método bateria de respuestas de la P11_2a a la P11_2k 
* Actualmente usuaria P11_4
* Método actual P11_6a, P11_6b, P11_6c, hay 3 respuestas xq se contesta Método más efectivo, DIU o NORPLANT y operación
* Lugar de obtención P11_7
* Usuaria actual (filtro) P11_21
* Estado Civil P11_23

* Empecemos con si es actualmente usuaria.
 
VALUE LABELS P11_4 1 'Si' 2 'No'.
Variable labels P11_4 'Usa actualmente algún método anticonceptivo'.
FREQUENCIES VARIABLES=P11_4
  /ORDER=ANALYSIS.

* En este caso la mayoría de los casos faltantes son porque en la pregunta P11_3 dicen que nunca usaría, entonces son 
respuestas por pase. 

FREQUENCIES VARIABLES=P11_3
  /ORDER=ANALYSIS.

* El estado civil de las mujeres

Value labels P11_23 1 'Union libre' 2 'Separada' 3 'Divorciada' 4 'Viuda' 5 'Casada' 6 'Soltera'.
FREQUENCIES VARIABLES=P11_23
  /ORDER=ANALYSIS.

* Recodifiquemosla para que sean unidas y no unidas solamente



RECODE
P11_23
(1=1) (5=1) (2=2) (3=2) (4=2) (6=2) into edo_civil.
VARIABLE LABELS edo_civil 'Estado Civil'.
Value labels edo_civil 1 'Unidas' 2 'No Unidas'.
EXECUTE.

FREQUENCIES VARIABLES=edo_civil
  /ORDER=ANALYSIS.

* Crucemos estado civil con si son usuarias.

CROSSTABS
  /TABLES=edo_civil BY P11_4
  /FORMAT=AVALUE TABLES
  /CELLS=COUNT ROW 
  /COUNT ROUND CELL.

* Ahora veamos el lugar de obtención de los metodos.  Lugar de obtención P11_7 para las usuarias. 

USE ALL.
COMPUTE filter_$=(P11_4 = 1).
VARIABLE LABEL filter_$ 'P3_8 >= 18 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMAT filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.

Variable labels P11_7 'Lugar de obtención del método'.
Value labels P11_7 1 'IMSS Hospital' 2 'IMSS Clinica' 3 'SSA Centro Salud' 4 'Encargado o auxiliar de comunidad' 5 'ISSSTE' 
6 'DIF' 7 'Otras de gobierno' 8 'MEXFAM' 9 'Farmacia' 10 'Consultorio, clínica u hospital privado' 11 'Tienda de autoservicio ' 12 
'Partera' 13 'Promotora' 
14 'Otro' 21 'Partera del IMSS' 23 'Partera de SSA' 25 'Partera del ISSSTE' 31 'Promotora del IMSS' 33 'Promotora de SSA' 35 
'Promotora ISSSTE' 36 'Promotora DIF' 99 'N.E.'.

FREQUENCIES VARIABLES=P11_7
  /ORDER=ANALYSIS.


